NGSPICE Simulation of CMOS Circuits

Wuqiong Zhao*

June 3, 2023

NGSPICE is a powerful open-source SPICE simulation software in command line, which can efficiently simulate CMOS circuits. Basic logic gates, including Not, Nand, And, Nor, are implemented and analyzed. The delay parameters and response plots can help understand the circuit characteristics. As examples, the 8-input NAND gate with three distinct designs is investigated, and the clock controlled SR latch is also simulated to design appropriate MOS parameters.

Contents

1.	Environments					
	1.1. Preparation	2				
	1.2. Settings	2				
	1.3. Development	2				
2 .	Definitions	2				
3.	Sources	3				
	3.1. FreePDK45	3				
	3.2. Inverter	3				
	3.3. NAND2	4				
	3.4. AND2	4				
	3.5. NOR2	5				
	3.6. AND8	6				
	3.6.1. Basic Components	6				
	3.6.2. AND8A (Symmetrical Design)	8				
	3.6.3. AND8B (NAND4A×2 + NOR2×1)	9				
	3.6.4. AND8C (AND4B×2 + AND2×1)	10				
	3.6.5. Analysis	11				
	3.7. Clock Controlled SR Latch	11				
	3.7.1. Schematic Design	11				
	3.7.2. MOS W/L Design	12				
	3.7.3. Simulation	12				
4.	License	12				

^{*} This is the course project of Fundamentals of VLSI Design, Southeast University, 2023 Spring. This document is available online at https://spice.tvj.one/report/NGSPICE_CMOS_Report.pdf.

The author is with the National Mobile Communications Research Laboratory, Southeast University, Nanjing 211189, China. (E-Mail: me@wqzhao.org)

2. Definitions 1. Environments

Re	ferences experiences	12
Α.	Resources A.1. Links	13
	A.1. Elliks	
В.	Source Code	13
	B.1. Inverter	13
	B.2. NAND2	15
	B.3. AND2	16
	B.4. NOR2	18
	B.5. NAND4	20
	B.6. AND4	21
	B.7. AND8	21
	B.7.1. Test Circuit	21
	B.7.2. AND8A	22
	B.7.3. AND8B	24
	B.7.4. AND8C	27
	B.8. Clock Controlled SR Latch	30
	B.9. MISC	32

1. Environments

1.1. Preparation

Install NGSPICE [1] CLI app. View the user manual [2] for more information.

1.2. Settings

NGSPICE is set to be compatible with HSPICE (see .spiceinit).

1.3. Development

My development environments:

- macOS 13 (Ventura) with M1 chip;
- NGSPICE 40 (Homebrew version).

There is *no* guarantee that the provided code can run on other platforms or other SPICE tools. Make changes if appropriate.

2. Definitions

Delay:

- t_r (tr): rise time (from output crossing $0.1V_{DD}$ to $0.9V_{DD}$);
- t_f (tf): fall time (from output crossing $0.9V_{DD}$ to $0.1V_{DD}$);
- t_{pdr} (tpdf): rising propagation delay (from input to rising output crossing $V_{DD}/2$);
- t_{pdf} (tpdf): falling propagation delay (from input to falling output crossing $V_{DD}/2$);
- t_{pd} (tpd): average propagation delay ($t_{pd} = (t_{pdr} + t_{pdf})/2$).

3. Sources 3. Sources

Operating corner:

• SS: slow-slow;

• MOM: nominal (average);

• FF: fast-fast.


3. Sources

3.1. FreePDK45

This is a 45 nm CMOS library. See README for more information. TNOM of FreePDK45 is 27C.

3.2. Inverter

The schematic diagram of the CMOS inverter is shown in Figure 1, with 1 PMOS and 1 NMOS. The design requirement is $t_r = t_f$ when $C_L = 24$ fF. The designed MOS parameters are shown in Table 1.



MOS	W	L
PMOS	360 nm	45 nm
NMOS	225 nm	45 nm

Figure 1: CMOS inverter schematic.

Table 1: Designed inverter MOS parameters.

Inverter Subcircuit

```
.subckt INV gnd i o vdd
  * src gate drain body type
  M1 vdd i o vdd PMOS_VTL W=360nm L=45nm
  M2 gnd i o gnd NMOS_VTL W=225nm L=45nm
.ends INV
```

Simulate with ngspice inv.cir, and the response of the inverter can be obtained as depicted in Figure 2.





Figure 2: Response of the inverter.

3.3. NAND2

The CMOS NAND2 gate is symmetrically designed with parameters for the worst case. The schematic is shown in Figure 3, where the value of W/L is labeled next to each MOS. The load capacitor is omitted. The designed parameters are shown in Table 2.



MOS	Number	W	L
PMOS	2	360 nm	45 nm
NMOS	2	225 nm	45 nm

Figure 3: CMOS NAND2 schematic.

Table 2: Designed NAND2 MOS parameters.

NAND2 Subcircuit

```
.subckt NAND2 gnd i1 i2 o vdd
         src
              gate drain body
                                type
     Mp1 vdd
                                PMOS_VTL W=360nm L=45nm
              i1
                          vdd
3
                    0
                                PMOS_VTL W=360nm L=45nm
     {\tt Mp2}\ vdd
              i2
                    0
                          vdd
     Mn1
         t1
               i1
                    0
                           gnd
                                NMOS_VTL W=450nm L=45nm
                                NMOS_VTL W=450nm L=45nm
     Mn2 gnd
              i2
                    t1
                          gnd
   .ends NAND2
```

The response simulated with ngspice nand2.cir is shown in Figure 4.




Figure 4: Response of the NAND2 gate.

3.4. AND2

AND2 is NAND2 + INV. The response of AND2 at the worst case is shown in Figure 5.




Figure 5: Response of the AND2 gate.

3. Sources 3.5. NOR2

3.5. NOR2

The schematic of the NOR2 gate is shown in Figure 6.




Figure 6: CMOS NOR2 schematic.

NOR2 Subcircuit

```
.subckt NOR2 gnd i1 i2 o vdd
         src
             gate drain body type
    Mp1 t1
                         vdd
                              PMOS_VTL W=720nm L=45nm
3
              i1
                   0
    Mp2 vdd
              i2
                   t1
                         vdd
                              PMOS_VTL W=720nm L=45nm
    Mn1 gnd
                              NMOS_VTL W=225nm L=45nm
              i1
                   0
                         gnd
                              NMOS_VTL W=225nm L=45nm
                         gnd
    Mn2 gnd
              i2
                   0
   .ends NOR2
```

The response of NOR2 gate is given in Figure 7 when simulating with ngspice nor2.cir.




Figure 7: Response of the AND2 gate.

3.6. AND8

8-input And gate. With a large fan-in, there can be several distinct designs. Here we want to investigate the performance of different designs.

The test circuit (defined in add8_test_inv2.inc) involves a 24 fF capacitor load at the output, and 8 sets of two stages of inverters for each input. For the inverter in the test circuit, NMOS has $W=0.75~\mu m$, $L=0.25~\mu m$, and PMOS has $W=2.60~\mu m$, $L=0.25~\mu m$.

The response simulation has the PVT condition of 1.0 V, FF, 25°C.

3.6.1. Basic Components

NAND4A This design directly extends the structure of NAND2 into NAND4. Its schematic is shown in Figure 8.




Figure 8: CMOS NAND4A schematic.

NAND8A This design directly extends the structure of NAND2 into NAND8. Its schematic is shown in Figure 9.




Figure 9: CMOS NAND8A schematic.

3.6.2. AND8A (Symmetrical Design)

This is the most basic case, extending 2-input NAND to 8-input NAND, before applying an inverter. The schematic of AND8A is shown in Figure 10. MOS parameters are specified in NAND8A.




Figure 10: CMOS AND8A schematic.

The simulation result with ngspice and8a.cir is shown in Table 3 and Figure 11.

Table 3: AND8A gate simulation result.

PVT Condition	t_r (ps)	t_f (ps)	t_{pdr} (ps)	t_{pdf} (ps)	P static (μW)	P dynamic (μW)
0.9 V, SS, 70°C	132.6	136.9	137.3	159.9	0.076	2.366
1.35 V, SS, 70°C	110.6	124.4	102.4	125.9	1.023	5.591
1.0 V, NOM, 25°C	90.5	99.2	83.8	110.5	0.227	2.733
1.5 V, NOM, 25°C	79.8	95.0	69.4	92.4	6.566	9.669
1.1 V, FF, 0°C	72.5	84.6	62.7	89.4	0.955	5.321
1.65 V, FF, 0°C	69.1	83.6	54.2	75.3	51.582	1.121




Figure 11: Response of the AND8A gate.

3.6.3. AND8B (NAND4A×2 + NOR2×1)

The schematic design using 2 NAND4A gates and 1 NOR gate is shown in Figure 12.




Figure 12: CMOS AND8B schematic.

The simulation result with ngspice and8b.cir is shown in Table 4 and Figure 13.

Table 4: AND8B gate simulation result.

PVT Condition	t_r (ps)	t_f (ps)	t_{pdr} (ps)	t_{pdf} (ps)	P static (μW)	P dynamic (μW)
0.9 V, SS, 70°C	118.8	131.7	102.8	106.2	0.030	2.233
1.35 V, SS, 70°C	96.2	113.1	78.3	83.3	0.641	6.204
1.0 V, NOM, 25°C	77.2	94.9	62.7	73.0	0.125	2.592
1.5 V, NOM, 25°C	65.7	85.5	52.0	61.0	4.895	12.301
1.1 V, FF, 0°C	59.4	79.9	46.2	58.8	0.510	5.289
1.65 V, FF, 0°C	53.3	74.3	30.5	50.3	43.767	11.248




Figure 13: Response of the AND8B gate.

3.6.4. AND8C (AND4B \times 2 + AND2 \times 1)

1.65 V, FF, 0°C

47.9

67.8

An AND4B gate is composed of 2 NAND gates and 1 NOR gate. The schematic design of AND8C using 2 NAND4A gates and 1 NOR gate is shown in Figure 14.




Figure 14: CMOS AND8C schematic.

The simulation result with ngspice and 8c.cir is shown in Table 5 and Figure 15.

T Condition	t_r (ps)	t_f (ps)	t_{pdr} (ps)	t_{pdf} (ps)	P static (μW)	P dyna
V, SS, 70°C	118.8	120.3	103.3	105.0	0.111	·
V SS 70°C	96.7	102.6	70.8	83.5	1 253	(

46.9

PVT namic (μW) $0.9 \, V$ 3.067 1.35 V, SS, 70°C 9.458 83.5 96.7 102.6 79.8 1.253 1.0 V, NOM, 25°C 81.5 86.9 68.4 73.3 0.301 4.164 1.5 V, NOM, 25°C 69.1 77.8 56.6 62.4 8.471 16.348 1.1 V, FF, 0°C 65.0 73.0 53.6 60.0 1.175 6.652

52.2

73.357

20.798

Table 5: AND8C gate simulation result.




Figure 15: Response of the AND8C gate.

3.6.5. Analysis

Among the three designs (AND8A, AND8B and AND8C), AND8B has the smallest latency, closely followed by AND8C, and the largest latency is observed with AND8A. Compared with AND8B, the fan-in of AND8B is significantly reduced, resulting in lower latency. Though AND8C has an even smaller fan-in, the number of stages in the circuit is larger than that of AND8B. Thus, AND8B achieves a reasonable tradeoff, and has the lowest latency.

There are also some other observations:

- A larger VDD can reduce the latency to some extent, but resulting in much larger power consumption.
- A faster operating corner (FF > NOM > SS) will help cut down latency, but also increases power.
- A higher temperature increases power in return for a reduced latency.

3.7. Clock Controlled SR Latch

3.7.1. Schematic Design

The schematic design of the clock controlled SR latch is shown in Figure 16, with 2 PMOS and 6 NMOS.




Figure 16: CMOS clock controlled SR latch schematic.

Clock Controlled SR Latch Subcircuit

```
.param WL = 5
    .subckt SR_LATCH_CLK gnd s r clk q qn vdd
         src
               gate drain body type
      M1 qn
                                   NMOS_VTL W=
                                                       90nm L=45nm
                      gnd
                             gnd
                q
      M2 qn
                             vdd
                                    {\tt PMOS\_VTL} \  \, {\tt W}{\tt =}
                                                      270nm L=45nm
                      vdd
5
                q
                                    {\tt NMOS\_VTL} \  \  {\tt W}{\tt =}
      M3 q
                qn
                      gnd
                             gnd
                                                       90nm L = 45nm
      M4 q
                      vdd
                             vdd
                                    PMOS_VTL W=
                                                      270nm L=45nm
                qn
                                    NMOS_VTL W = \{WL * 45nm\} L=45nm
      M5 ts
                             gnd
8
                S
                      gnd
                                    NMOS_VTL W = \{WL * 45nm\} L=45nm
      M6 qn
                clk
                      ts
                             gnd
                                    NMOS_VTL W = \{WL * 45nm\} L=45nm
10
      M7 tr
                r
                      gnd
                             gnd
                clk
                                    NMOS_VTL W = \{WL * 45nm\} L=45nm
      M8 q
                      tr
                             gnd
    .ends SR_LATCH_CLK
```

You need to specify the parameter WL, for example .param WL = 5.

References 4. License

3.7.2. MOS W/L Design

We need to determine the appropriate W/L for M_5 to M_8 . Using a sweep, implemented by alterparam within a foreach loop, we can obtain Figure 17.




Figure 17: Clock controlled SR latch with different W/L values.

Clearly, we need W/L > 4.5 (at least 4.2) for the latch to work properly. $(M_1/M_3 \text{ and } M_2/M_4 \text{ have } W/L \text{ as 2 and 6, respectively.})$

3.7.3. Simulation

The response of NOR2 gate is given in Figure 18 when simulating with ngspice SR_latch_clk.cir.




Figure 18: Response of the clock controlled SR latch when W/L = 6.

4. License

Copyright © 2023 Wuqiong Zhao (me@wqzhao.org). This project is distributed by an MIT License.

References

- [1] H. Vogt, A. Giles, P. Nenzi, and D. Warning, *NGSPICE 40 open source spice simulator*, 2023. [Online]. Available: https://ngspice.sourceforge.io/.
- [2] H. Vogt, A. Giles, P. Nenzi, and D. Warning, *Ngspice user's manual version 40 (ngspice release version)*, Apr. 2023. [Online]. Available: https://ngspice.sourceforge.io/docs/ngspice-40-manual.pdf.

B. Source Code A. Resources

A. Resources

A.1. Links

• GitHub: Teddy-van-Jerry/ngspice-cmos

• Website: spice.tvj.one

A.2. Schematic Graph

All schematic graphs in this document is drawn with TikZ using LAT_EX, you can find the source code in the GitHub repository.

B. Source Code

B.1. Inverter

Listing 5: inv.inc

```
* ------
  * Circuit : CMOS Inverter
  * Description: 1 PMOS + 1 NMOS
  * Author
             : Wuqiong Zhao (me@wqzhao.org)
5
  * Date
            : 2023-06-02
  * License
             : MIT
  * -----
8
  .subckt INV gnd i o vdd
10
11
    * src gate drain body type
    M1 vdd i o vdd PMOS_VTL W=360nm L=45nm
12
13
   M2 gnd i
               0
                    gnd NMOS_VTL W=225nm L=45nm
14
   .ends INV
15
  .subckt NOT gnd i o vdd
16
   * src gate drain body type
M1 vdd i o vdd PMOS_VTL W=360nm L=45nm
M2 gnd i o gnd NMOS_VTL W=225nm L=45nm
17
18
19
  .ends NOT
```

Listing 6: inv.cir

```
* ------
  * Circuit : CMOS Inverter with 1 PMOS + 1 NMOS
  * Description: tr = tf when C_L = 0.024pF
  * Author
           : Wuqiong Zhao (me@wqzhao.org)
5
  6
  * ------
  * Reference:
10
  * https://github.com/cornell-ece5745/ece5745-tut10-spice/blob/master/sim/inv-sim.sp
11
  .title CMOS Inverter
13
14
  * Parameters and Model
15
  .param VDD='1.0V'
17
18
  .temp 27
  .inc ./FreePDK45/ff.inc
19
20
  * Supply Voltage Source
21
  Vdd vdd gnd VDD
23
```

B. Source Code B.1. Inverter

```
* Inverter
25
26
27
   .inc ./inv.inc
   XInv gnd in out vdd INV
28
   * Load Capacitor
30
31
32
   CL out gnd 24fF
33
   * Input Signals
34
35
   Vin in gnd PWL
36
37
   + (
       0.0ns 0V
38
       0.9ns 0V
39
      1.1ns VDD
       1.9ns VDD
41
      2.1ns 0V
42
43
       3.0ns 0V
   + )
44
45
   * Analysis
46
47
   .ic V(out)=VDD
48
   .tran 0.005ns 3ns
49
50
51
   .control
52
     run
53
     * >>>> plot >>>>>
     set xgridwidth = 2
54
     set xbrushwidth = 3
55
     * "svgwidth", "svgheight", "svgfont-size", "svgfont-width", "svguse-color",

→ "svgstroke-width", "svggrid-width",
     set svg_intopts = ( 1024 256 16 0 1 2 0 )
57
     * "svgbackground", "svgfont-family", "svgfont"
58
     setcs svg_stropts = ( white Arial Arial )
59
60
     set hcopydevtype = svg
                   = black
61
     set color1
     set color2
                       = blue
62
63
     set color3
                       = red
64
65
     hardcopy fig/plot_inv_t.svg
66
     + out in
     + title 'CMOS Inverter'
67
     + xlabel 't'
+ ylabel 'Voltage'
68
69
     + ylimit 0 1
70
71
72
     * for MS Windows, using Edge
     if $oscompiled = 1 | $oscompiled = 8
73
      shell Start fig/plot_inv_t.svg
74
75
     else
76
       if $oscompiled = 7
          * macOS (using Safari, no need to install X11)
77
78
         shell open -a safari fig/plot_inv_t.svg &
79
          * for CYGWIN, Linux, using feh and X11
80
         shell feh --magick-timeout 1 fig/plot_inv_t.svg &
81
82
     end
83
84
     * <<<<< plot <<<<<
85
   .endc
86
   * Measurement
88
   .measure tran tr trig V(out) val='VDD*0.1' rise=1 targ V(out) val='VDD*0.9' rise=1
89
   .measure tran tf trig V(out) val='VDD*0.9' fall=1 targ V(out) val='VDD*0.1' fall=1
   .measure tran tpdr trig V(in) val='VDD/2' fall=1 targ V(out) val='VDD/2'
.measure tran tpdf trig V(in) val='VDD/2' rise=1 targ V(out) val='VDD/2'
                                                                                      rise=1
91
                                                   rise=1 targ V(out) val='VDD/2'
92
   .measure tran tpd param='(tpdr+tpdf)/2'
93
94
95
```

B.2. NAND2

Listing 7: nand2.inc

```
* -----
    * Circuit : CMOS NAND2 Gate
2
    * Description: 2 PMOS + 2 NMOS
                     : Wuqiong Zhao (me@wqzhao.org)
    * Author
5
    * Date : 2023-06-01
* License : MIT
    * -----
8
    .subckt NAND2 gnd i1 i2 o vdd
10
11
       * src gate drain body type

        Mp1 vdd
        i1 o
        vdd
        PMOS_VTL
        W=360nm
        L=45nm

        Mp2 vdd
        i2 o
        vdd
        PMOS_VTL
        W=360nm
        L=45nm

12
13
       Mn1 t1 i1 o
                                   gnd NMOS_VTL W=450nm L=45nm
      Mn2 gnd i2 t1 gnd NMOS_VTL W=450nm L=45nm
15
     .ends NAND2
16
17
18
     .subckt NAND gnd i1 i2 o vdd
19
       * src gate drain body type
       Mp1 vdd i1 o vdd PMOS_VTL W=360nm L=45nm
20

        Mp2
        vdd
        i2
        o
        vdd
        PMOS_VTL
        W=360nm
        L=45nm

        Mn1
        t1
        i1
        o
        gnd
        NMOS_VTL
        W=450nm
        L=45nm

        Mn2
        gnd
        i2
        t1
        gnd
        NMOS_VTL
        W=450nm
        L=45nm

21
      Mn1 t1 i1 o
Mn2 gnd i2 t1
22
23
    .ends NAND
```

Listing 8: nand2.cir

```
* ------
  * Circuit : CMOS NAND2 Gate
2
  \star Description: 2 PMOS + 2 NMOS
  * Author : Wuqiong Zhao (me@wqzhao.org)
* Date : 2023-06-01
* License : MIT
5
6
  * ------
8
  .title CMOS NAND2
10
11
12
  * Parameters and Model
13
14
  .param VDD='1.0V'
15
  .temp 27
  .inc ./FreePDK45/ff.inc
16
17
  * Supply Voltage Source
18
19
  Vdd vdd gnd VDD
21
  * NAND2
22
23
  XNAND2 gnd in1 in2 out vdd NAND2
24
25
   .inc ./nand2.inc
26
  * Load Capacitor
27
28
  CL out gnd 24fF
29
30
  * Input Signals
31
32
  Vin1 in1 gnd PWL
33
34
  + (
      0.0ns 0V
35
  + 0.9ns 0V
37
  +
      1.1ns VDD
     3.0ns VDD
38
  +
  + )
40
```

```
Vin2 in2 gnd PWL
41
42
        0.0ns 0V
43
        0.9ns 0V
44
45
        1.1ns VDD
       1.9ns VDD
46
47
       2.1ns 0V
48
       3.0ns 0V
   + )
49
50
   * Analysis
51
52
   .ic V(out)=VDD
   .tran 0.005ns 3ns
54
55
57
     run
58
     * >>>> plot >>>>>
     set xgridwidth = 2
59
     set xbrushwidth = 3
60
      * "svgwidth", "svgheight", "svgfont-size", "svgfont-width", "svguse-color",
61

→ "svgstroke-width", "svggrid-width",
     set svg_intopts = ( 1024 256 16 0 1 2 0 )
* "svgbackground", "svgfont-family", "svgfont"
62
63
      setcs svg_stropts = ( white Arial Arial )
64
65
      set hcopydevtype = svg
                    = black
= red
66
      set color1
      set color2
67
68
      set color3
                       = blue
     set color4
69
                       = green
70
     hardcopy fig/plot_nand2_t.svg
     + in1 out in2
72
      + title 'CMOS NAND2'
73
     + xlabel 't'
74
     + ylabel 'Voltage'
75
76
      + ylimit 0 1
77
      \ensuremath{^{*}} for MS Windows, using Edge
78
79
      if $oscompiled = 1 | $oscompiled = 8
       shell Start fig/plot_nand2_t.svg
80
81
      else
82
       if $oscompiled = 7
           * macOS (using Safari, no need to install X11)
83
84
          shell open -a safari fig/plot_nand2_t.svg &
85
         * for CYGWIN, Linux, using feh and X11
86
          shell feh --magick-timeout 1 fig/plot_nand2_t.svg &
88
        end
89
     end
     * <<<<< plot <<<<
91
    .endc
92
   * Measurement
93
94
    .measure tran tr    trig V(out) val='VDD*0.1' rise=1 targ V(out) val='VDD*0.9' rise=1
95
   .measure tran tf trig V(out) val='VDD*0.9' fall=1 targ V(out) val='VDD*0.1' fall=1
96
   .measure tran tpdr trig V(in2) val='VDD/2' fall=1 targ V(out) val='VDD/2'
    .measure tran tpdf trig V(in1) val='VDD/2'
                                                   rise=1 targ V(out) val='VDD/2'
   .measure tran tpd param='(tpdr+tpdf)/2'
99
100
101
   . end
```

B.3. AND2

Listing 9: and2.inc

```
4
           : Wuqiong Zhao (me@wqzhao.org)
: 2023-06-02
: MIT
  * Author
  * Date
6
  * License
  * ------
  .subckt AND2 gnd i1 i2 o vdd
10
11
    XNAND gnd i1 i2 o1 vdd NAND2
   XInv gnd o1
                 o vdd INV
12
  .ends AND2
13
14
  .subckt AND gnd i1 i2 o vdd
15
   XNAND gnd i1 i2 o1 vdd NAND2
   XInv gnd o1 o vdd INV
17
  .ends AND
18
19
.inc ./nand2.inc
  .inc ./inv.inc
21
```

Listing 10: and2.cir

```
* -----
  * Circuit : CMOS AND2 Gate
2
3
  * Description: NMOS2 + Inverter (3 PMOS + 3 NMOS)
  * Author : Wuqiong Zhao (me@wqzhao.org)
* Date : 2023-06-02
* License : MIT
5
  * Date
  * ______
8
  .title CMOS AND2
10
11
  * Parameters and Model
12
13
   .param VDD='1.0V'
14
  .temp 27
15
  .inc ./FreePDK45/ff.inc
16
  * Supply Voltage Source
18
19
20
  Vdd vdd gnd VDD
21
22
  * AND2
23
24 XAND2 gnd in1 in2 out vdd AND2
  .inc ./and2.inc
26
  * Load Capacitor
27
  CL out gnd 24fF
29
30
  * Input Signals
31
32
33
  Vin1 in1 gnd PWL
34
  + (
     0.0ns 0V
35
36
     0.9ns 0V
     1.1ns VDD
37
     3.0ns VDD
38
39
  + )
40
41
  Vin2 in2 gnd PWL
42
  + (
     0.0ns 0V
43
      0.9ns 0V
      1.1ns VDD
45
     1.9ns VDD
46
47
  + 2.1ns 0V
     3.0ns 0V
48
49
  + )
51 * Analysis
```

B. Source Code B.4. NOR2

```
52
    .ic V(out)=0
53
   .tran 0.005ns 3ns
55
56
    .control
57
      * >>>> plot >>>>>
58
59
      set xgridwidth = 2
     set xbrushwidth = 3
60
      * "svgwidth", "svgheight", "svgfont-size", "svgfont-width", "svguse-color",
61

    "svgstroke-width", "svggrid-width"

      set svg_intopts = ( 1024 256 16 0 1 2 0 )
62
      * "svgbackground", "svgfont-family", "svgfont"
63
      setcs svg_stropts = ( white Arial Arial )
64
65
      set hcopydevtype = svg
                     = black
      set color1
67
     set color2
                       = red
68
      set color3
                       = blue
69
     set color4
                      = green
70
71
     hardcopy fig/plot_and2_t.svg
     + in1 out in2
72
     + title 'CMOS AND2'
73
     + xlabel 't'
74
     + ylabel 'Voltage'
75
     + ylimit 0 1
76
77
      \ensuremath{^{*}} for MS Windows, using Edge
78
79
     if $oscompiled = 1 | $oscompiled = 8
       shell Start fig/plot_and2_t.svg
80
81
      else
       if $oscompiled = 7
          * macOS (using Safari, no need to install X11)
83
84
          shell open -a safari fig/plot_and2_t.svg &
85
           * for CYGWIN, Linux, using feh and X11
86
87
          shell feh --magick-timeout 1 fig/plot_and2_t.svg &
88
        end
89
      end
90
      * <<<<< plot <<<<
   .endc
91
92
93
   * Measurement
94
   .measure tran tr trig V(out) val='VDD*0.1' rise=1 targ V(out) val='VDD*0.9' rise=1
    .measure tran tf
                      trig V(out) val='VDD*0.9' fall=1 targ V(out) val='VDD*0.1' fall=1
96
    .measure tran tpdr trig V(in1) val='VDD/2'
                                                  rise=1 targ V(out) val='VDD/2'
97
                                                                                     rise=1
   .measure tran tpdf trig V(in2) val='VDD/2'
                                                   fall=1 targ V(out) val='VDD/2'
   .measure tran tpd param='(tpdr+tpdf)/2'
99
100
   .end
```

B.4. NOR2

Listing 11: nor2.inc

```
* ------
   * Circuit : CMOS NOR2 Gate
  * Description: 2 PMOS + 2 NMOS
3
4
              : Wuqiong Zhao (me@wqzhao.org)
  * Date
             : 2023-06-01
6
  * License
            : MIT
   * ------
8
9
10
   .subckt NOR2 gnd i1 i2 o vdd
    * src gate drain body type
11
                     vdd PMOS_VTL W=720nm L=45nm
12
    Mp1 t1
            i1 o
    Mp2 vdd i2 t1
                     vdd PMOS_VTL W=720nm L=45nm
    \label{eq:mn1} \mbox{Mn1 gnd} \quad \mbox{i1} \quad \mbox{o} \qquad \mbox{gnd} \quad \mbox{NMOS\_VTL } \mbox{$W$=$225nm} \ \mbox{$L$=$45nm}
```

B. Source Code B.4. NOR2

```
Mn2 gnd i2 o gnd NMOS_VTL W=225nm L=45nm
15
16
   .ends NOR2
17
   .subckt NOR gnd i1 i2 o vdd
18
19
     * src gate drain body type
              i1 o vdd PMOS_VTL W=720nm L=45nm
     Mp1 t1
20
     Mp2 vdd i2 t1 vdd PMOS_VTL W=720nm L=45nm
21
    Mn1 gnd i1 o
Mn2 gnd i2 o
                          gnd NMOS_VTL W=225nm L=45nm gnd NMOS_VTL W=225nm L=45nm
22
23
   .ends NOR
```

Listing 12: nor2.cir

```
* -----
  * Circuit : CMOS NOR2 Gate
2
  * Description: 2 PMOS + 2 NMOS
3
  * Author : Wuqiong Zhao (me@wqzhao.org)
* Date : 2023-06-02
* License : MIT
  * Author
5
6
  * -----
8
  .title CMOS NOR2
10
11
  * Parameters and Model
12
13
14
  .param VDD='1.0V'
15
   .temp 27
  .inc ./FreePDK45/ff.inc
16
17
  * Supply Voltage Source
18
19
20 Vdd vdd gnd VDD
21
  * NOR2
22
23
  XNOR2 gnd in1 in2 out vdd NOR2
24
25
   .inc ./nor2.inc
26
  * Load Capacitor
27
28
29 CL out gnd 24fF
30
  * Input Signals
31
32
33
  Vin1 in1 gnd PWL
  + (
34
     0.0ns VDD
35
36
  + 0.9ns VDD
      1.1ns 0V
37
  +
      3.0ns 0V
38
  +
  + )
39
40
41
  Vin2 in2 gnd PWL
42
  + (
      0.0ns VDD
43
44
      0.9ns VDD
     1.1ns 0V
45
     1.9ns 0V
46
47
      2.1ns VDD
     3.0ns VDD
48
49 + )
50
  * Analysis
51
  .ic V(out)=0V
.tran 0.005ns 3ns
53
54
55
56 .control
  run
* >>>> plot >>>>>
57
58
set xgridwidth = 2
```

```
set xbrushwidth = 3
60
      * "svgwidth", "svgheight", "svgfont-size", "svgfont-width", "svguse-color",
61

→ "svgstroke-width", "svggrid-width",
      set svg_intopts = ( 1024 256 16 0 1 2 0 )
* "svgbackground", "svgfont-family", "svgfont"
62
63
      setcs svg_stropts = ( white Arial Arial )
64
65
      set hcopydevtype = svg
                    = black
66
      set color1
      set color2
                       = red
67
      set color3
                       = blue
68
      set color4
                       = green
69
70
     hardcopy fig/plot_nor2_t.svg
71
      + in1 out in2
72
      + title 'CMOS NOR2'
73
      + xlabel 't'
      + ylabel 'Voltage'
75
76
      + ylimit 0 1
77
      * for MS Windows, using Edge
78
79
      if $oscompiled = 1 | $oscompiled = 8
       shell Start fig/plot_nor2_t.svg
80
81
      else
82
        if $oscompiled = 7
          * macOS (using Safari, no need to install X11)
83
84
          shell open -a safari fig/plot_nor2_t.svg &
85
           * for CYGWIN, Linux, using feh and X11
86
87
          shell feh --magick-timeout 1 fig/plot_nor2_t.svg &
88
        end
89
     end
     * <<<<< plot <<<<<
91
    .endc
92
   * Measurement
93
94
    .measure tran tr    trig V(out) val='VDD*0.1' rise=1 targ V(out) val='VDD*0.9' rise=1
95
    .measure tran tf trig V(out) val='VDD*0.9' fall=1 targ V(out) val='VDD*0.1' fall=1
96
   .measure tran tpdr trig V(in2) val='VDD/2'
                                                   fall=1 targ V(out) val='VDD/2'
97
                                                                                       rise=1
    .measure tran tpdf trig V(in2) val='VDD/2'
                                                   rise=1 targ V(out) val='VDD/2'
   .measure tran tpd param='(tpdr+tpdf)/2'
99
100
    .end
```

B.5. NAND4

Listing 13: nand4a.inc

```
* -----
  * Circuit : CMOS NAND4 Gate Type
  * Description: 4 PMOS + 4 NMOS (Symmetrical Design)
3
  * Author
            : Wuqiong Zhao (me@wqzhao.org)
  * Date
             : 2023-06-02
6
  * License
            : MTT
  * ------
9
  .subckt NAND4A gnd i1 i2 i3 i4 o vdd
10
    * src gate drain body type
11
                    vdd PMOS_VTL W=360nm L=45nm
    Mp1 vdd i1 o
12
    Mp2 vdd
                    vdd PMOS_VTL W=360nm L=45nm
13
           i2
               0
                    vdd PMOS_VTL W=360nm L=45nm
    Mp3 vdd i3
14
               0
15
    Mp4 vdd i4
               О
                    vdd PMOS_VTL W=360nm L=45nm
16
    Mn1 t1
           i1
               0
                    gnd
                        NMOS_VTL W=900nm L=45nm
                    gnd NMOS_VTL W=900nm L=45nm
    Mn2 t2
17
           i2
               t1
18
    Mn3 t3
           i3
               t2
                    gnd NMOS_VTL W=900nm L=45nm
    Mn4 gnd
           i4
               t3
                    gnd NMOS_VTL W=900nm L=45nm
19
  .ends NAND4A
20
```

B.6. AND4

Listing 14: and4b.inc

```
* ------
  * Circuit : CMOS AND4 Gate Type B
2
  * Description: NAND2 * 2 + NOR * 1
  * Author
5
           : Wuqiong Zhao (me@wqzhao.org)
           : 2023-06-02
  * License : MIT
  * -----
8
  .subckt AND4B gnd i1 i2 i3 i4 o vdd
10
   * src gate drain body type
11
   XNAND2_1 gnd i1 i2 t1 vdd NAND2
12
   XNAND2_2 gnd i3 i4 t2 vdd NAND2
13
   XNOR gnd t1 t2 o vdd NOR2
  .ends AND4B
15
16
17
  .inc ./nand2.inc
18
  .inc ./nor2.inc
```

B.7. AND8

B.7.1. Test Circuit

Listing 15: and8_test_inv2.inc

```
* -----
  * Circuit : Test Circuit With Capicitor Load
  * Description: 2 Inverters at the Input
3
  * Author
              : Wuqiong Zhao (me@wqzhao.org)
  * Date
              : 2023-06-02
6
   * License : MIT
  * ------
9
10
   .subckt INV2_TEST gnd
                  i1 i2 i3 i4 i5 i6 i7 i8
11
                   t1 t2 t3 t4 t5 t6 t7 t8
12
  +
13
                   out vdd
            gnd i o vdd
14
    XInv_ss1 gnd i1 t1 vdd INV_SS
15
16
    XInv_ss2 gnd i2 t2 vdd INV_SS
    XInv_ss3 gnd i3 t3 vdd INV_SS
17
18
    XInv_ss4 gnd i4 t4 vdd INV_SS
    XInv_ss5 gnd i5 t5 vdd INV_SS
19
    XInv_ss6 gnd i6 t6 vdd INV_SS
20
21
    XInv_ss7 gnd i7 t7 vdd INV_SS
22
    XInv_ss8 gnd i8 t8 vdd INV_SS
     * Load
23
    CL out gnd 24fF
   .ends INV2_TEST
25
26
  .subckt INV_SS gnd i o vdd
27
   XInv_s1 gnd i t vdd INV_S
28
29
    XInv_s2 gnd t o vdd INV_S
30
31
   * Inverter used in the test circuit
32
   .subckt INV_S gnd i o vdd
33
    * src gate drain body type
34
                  vdd PMOS_VTL W=0.75um L=0.25um gnd NMOS_VTL W=2.60um L=0.25um
35
    M1 vdd i o
    M2 gnd i
36
                0
  .ends INV_S
```

Listing 16: and8_test_pow.inc

```
* ------
  * Script : Measure AND8 Gate Static and Dynamic Power
  * Description: Measure the power for device 'xand8'.
3
  * Author
             : Wuqiong Zhao (me@wqzhao.org)
            : 2023-06-02
: MIT
  * Date
6
  * License
  * ------
8
9
10
   .probe P(XAND8)
  .control
11
12
    * Total power
    let p_total = mean(xand8:power)
13
    * Static power (average of high and low)
14
    let p_static =
15
16
    + (
    + mean(xand8:power[length(xand8:power) / 6, length(xand8:power) / 3]) +
17
18
    + mean(xand8:power[length(xand8:power) / 2, length(xand8:power) / 1.5])
19
    + ) / 2
    * Dynamic power (difference of total power and static power)
20
    let p_dynamic = p_total - p_static
    print p_total
22
23
    print p_static
    print p_dynamic
24
25
  .endc
```

B.7.2. AND8A

Listing 17: and8a.inc

```
* ------
  * Circuit : CMOS AND8 Gate Type A
2
  ^{*} Description: 8 PMOS + 8 NMOS (Symmetrical Design) + 1 Inv
3
  * Author
            : Wuqiong Zhao (me@wqzhao.org)
5
  * Date
           : 2023-06-02
6
  * License
           : MIT
  * ------
8
  .subckt AND8A gnd i1 i2 i3 i4 i5 i6 i7 i8 o vdd
10
   XNAND8a gnd i1 i2 i3 i4 i5 i6 i7 i8 o_inv vdd NAND8A
11
   Xinv gnd o_inv o vdd INV
12
  .ends AND8A
13
14
  .inc ./nand8a.inc
15
  .inc ./inv.inc
16
```

Listing 18: and8a.cir

```
*
1
  * Circuit : CMOS AND8 Gate Type A
  * Description: 8 PMOS + 8 NMOS (Symmetrical Design) + 1 Inv
3
5
           : Wuqiong Zhao (me@wqzhao.org)
  * Date
          : 2023-06-02
6
  * License
          : MIT
  * ------
8
  .title CMOS AND8A
10
11
  * Parameters and Model
12
13
  .param VDD='1.0V'
14
15
  .temp 25
  .inc ./FreePDK45/ff.inc
16
17
 * Supply Voltage Source
```

```
19
20
   Vdd vdd gnd VDD
21
   * AND8A
22
23
   XAND8 gnd in1 in2 in3 in4 in5 in6 in7 in8 out vdd AND8A
24
   .inc ./and8a.inc
25
26
   * Test Circuit
27
28
29
   XTest gnd
   + ii1 ii2 ii3 ii4 ii5 ii6 ii7 ii8
30
        in1 in2 in3 in4 in5 in6 in7 in8
   + out vdd
.inc ./and8_test_inv2.inc
                                INV2_TEST
32
33
   * Input Signals
35
36
Vii1 ii1 gnd PWL
38 + (
       0.0ns 0V
39
   + 0.9ns 0V
40
   + 1.1ns VDD
+ 3.0ns VDD
+ )
41
42
43
45
   Vii2 ii2 gnd PWL
   + (
46
   + 0.0ns 0V
+ 0.9ns 0V
+ 1.1ns VDD
47
48
49
50 + 3.0ns VDD
51
   + )
52
53 Vii3 ii3 gnd PWL
54 + (
       0.0ns 0V
55
        0.9ns 0V
56
   + 1.1ns VDD
57
58
        3.0ns VDD
59 + )
60
61
   Vii4 ii4 gnd PWL
62 + (
63 + 0.0ns 0V
   + 0.9ns 0V
+ 1.1ns VDD
64
65
66 + 3.0ns VDD
67
68
69 Vii5 ii5 gnd PWL
70 + (
      0.0ns 0V
71
   + 0.9ns 0V
72
   + 1.1ns VDD
+ 3.0ns VDD
73
74
75 + )
76
77
   Vii6 ii6 gnd PWL
   + (
78
   + 0.0ns 0V
79
       0.9ns 0V
80
      1.1ns VDD
81
82 + 3.0ns VDD
83 + )
84
85 Vii7 ii7 gnd PWL
86 + (
   + 0.0ns 0V
87
88 + 0.9ns 0V
89 + 1.1ns VDD
90 + 3.0ns VDD
91 + )
```

```
92
93
    Vii8 ii8 gnd PWL
    + (
         0.0ns 0V
95
96
         0.9ns 0V
        1.1ns VDD
97
        1.9ns VDD
98
99
         2.1ns 0V
        3.0ns 0V
100
101
    + )
102
    * Analysis
103
104
    .ic
            V(out)=0
105
    .tran 0.005ns 3ns
106
108
    .control
109
      * >>>> plot >>>>>
110
      set xgridwidth = 2
111
112
      set xbrushwidth = 3
       * "svgwidth", "svgheight", "svgfont-size", "svgfont-width", "svguse-color",
113
       set svg_intopts = ( 1024 256 16 0 1 2 0 )
      * "svgbackground", "svgfont-family", "svgfont"
115
      setcs svg_stropts = ( white Arial Arial )
116
117
      set hcopydevtype = svg
                      = black
      set color1
118
119
      set color2
                         = red
      set color3
                         = blue
120
                         = green
      set color4
121
      hardcopy fig/plot_and8a_t.svg
123
124
      + in1 out in8
      + title 'CMOS AND8a'
125
      + xlabel 't'
126
      + ylabel 'Voltage'
127
      + ylimit -0.2 1.2
128
      + ydelta 0.5
129
130
      ^{*} for MS Windows, using Edge
131
132
      if $oscompiled = 1 | $oscompiled = 8
        shell Start fig/plot_and8a_t.svg
133
      else
134
135
        if $oscompiled = 7
           * macOS (using Safari, no need to install X11)
136
           shell open -a safari fig/plot_and8a_t.svg &
137
138
           * for CYGWIN, Linux, using feh and X11
139
          shell feh --magick-timeout 1 fig/plot_and8a_t.svg &
140
      end
142
      * <<<<< plot <<<<
143
    .endc
144
145
146
    * Measurement
147
    .measure tran tr trig V(out) val='VDD*0.1' rise=1 targ V(out) val='VDD*0.9' rise=1 .measure tran tf trig V(out) val='VDD*0.9' fall=1 targ V(out) val='VDD*0.1' fall=1 .measure tran tpdr trig V(in1) val='VDD/2' rise=1 targ V(out) val='VDD/2' rise=1
148
150
    .measure tran tpdf trig V(in8) val='VDD/2'
                                                       fall=1 targ V(out) val='VDD/2'
151
152
    .measure tran tpd param='(tpdr+tpdf)/2
153
    * power dissipation of the AND8 gate
155
    .inc ./and8_test_pow.inc
156
```

B.7.3. AND8B

Listing 19: and8b.inc

```
* ______
2 * Circuit : CMOS AND8 Gate Type B
  * Description: NAND4A * 2 + NOR2 *
3
  * Author
           : Wuqiong Zhao (me@wqzhao.org)
  * Date : 2023-06-02
* License : MIT
  * Date
6
  * ------
9
  .subckt AND8B gnd i1 i2 i3 i4 i5 i6 i7 i8 o vdd
    * src gate drain body type
11
    XNAND4A_1 gnd i1 i2 i3 i4 t1 vdd NAND4A
12
13
    XNAND4A_2 gnd i5 i6 i7 i8 t2 vdd NAND4A
   XNOR2 gnd t1 t2
                     o vdd NOR2
14
  .ends AND8B
15
16
  .inc ./nand4a.inc
17
18 .inc ./nor2.inc
```

Listing 20: and8b.cir

```
* ------
  * Circuit : CMOS AND8 Gate Type B
  * Description: NAND4A * 2 + NOR2 *
3
  * Author
              : Wuqiong Zhao (me@wqzhao.org)
: 2023-06-02
5
   * Date
6
  * License : MIT
8
  .title CMOS AND8B
10
11
  * Parameters and Model
12
13
   .param VDD='1.0V'
14
15
   .temp 25
  .inc ./FreePDK45/ff.inc
16
17
  * Supply Voltage Source
18
19
20 Vdd vdd gnd VDD
21
   * AND8B
22
23
   XAND8 gnd in1 in2 in3 in4 in5 in6 in7 in8 out vdd AND8B
24
25
   .inc ./and8b.inc
   * Test Circuit
27
28
29 XTest gnd
        ii1 ii2 ii3 ii4 ii5 ii6 ii7 ii8
30
31
        in1 in2 in3 in4 in5 in6 in7 in8
        out vdd
                             INV2_TEST
33
   .inc ./and8_test_inv2.inc
  * Input Signals
35
36
   Vii1 ii1 gnd PWL
37
38
  + (
  + 0.0ns 0V
39
      0.9ns 0V
40
     1.1ns VDD
41
     3.0ns VDD
  + )
43
44
45 Vii2 ii2 gnd PWL
46
  + (
47
      0.0ns 0V
      0.9ns 0V
48 +
     1.1ns VDD
49 +
  + 3.0ns VDD
```

```
+ )
51
52
Vii3 ii3 gnd PWL
54 + (
       0.0ns 0V
55
        0.9ns 0V
56
   + 1.1ns VDD
57
58
       3.0ns VDD
59
60
    Vii4 ii4 gnd PWL
61
62 + (
63 + 0.0ns 0V
64
       0.9ns 0V
       1.1ns VDD
65
66 + 3.0ns VDD
67 + )
68
69 Vii5 ii5 gnd PWL
70 + (
       0.0ns 0V
71
   + 0.9ns 0V
72
73 + 1.1ns VDD
74 + 3.0ns VDD
75 + )
76
77
   Vii6 ii6 gnd PWL
78
   + (
79 + 0.0ns 0V
   + 0.9ns 0V
+ 1.1ns VDD
80
81
   + 3.0ns VDD
   + )
83
84
85 Vii7 ii7 gnd PWL
86 + (
       0.0ns 0V
87
        0.9ns 0V
88
   + 1.1ns VDD
89
90
        3.0ns VDD
91 + )
92
93
   Vii8 ii8 gnd PWL
94 + (
95 + 0.0ns 0V
96
   +
       0.9ns 0V
       1.1ns VDD
97
       1.9ns VDD
        2.1ns 0V
99
    +
        3.0ns 0V
100
   +
   + )
101
102
   * Analysis
103
104
    .ic V(out)=0
105
    .tran 0.005ns 3ns
106
107
108
    .control
    run
109
      * >>>> plot >>>>>
110
     set xgridwidth = 2
111
      set xbrushwidth = 3
112
      * "svgwidth", "svgheight", "svgfont-size", "svgfont-width", "svguse-color",

→ "svgstroke-width", "svggrid-width",
113
      set svg_intopts = ( 1024 256 16 0 1 2 0 )
114
      * "svgbackground", "svgfont-family", "svgfont"
115
116
      setcs svg_stropts = ( white Arial Arial )
      set hcopydevtype = svg
117
                    = black
= red
118
      set color1
      set color2
119
      set color3
                       = blue
120
121
      set color4
                       = green
122
```

```
hardcopy fig/plot_and8b_t.svg
123
124
      + in1 out in8
      + title 'CMOS AND8b'
125
      + xlabel 't'
126
      + ylabel 'Voltage'
127
      + ylimit -0.2 1.2
128
129
      + ydelta 0.5
      ^{\star} for MS Windows, using Edge
131
132
      if $oscompiled = 1 | $oscompiled = 8
133
         shell Start fig/plot_and8b_t.svg
134
      else
135
        if $oscompiled = 7
           * macOS (using Safari, no need to install X11)
136
137
           shell open -a safari fig/plot_and8b_t.svg &
            * for CYGWIN, Linux, using feh and X11
139
140
           shell feh --magick-timeout 1 fig/plot_and8b_t.svg &
141
         end
142
      end
143
      * <<<<< plot <<<<
144
    .endc
145
146
    * Measurement
147
148
    .measure tran tr trig V(out) val='VDD*0.1' rise=1 targ V(out) val='VDD*0.9' rise=1
                         trig V(out) val='VDD*0.9' fall=1 targ V(out) val='VDD*0.1' fall=1
trig V(in1) val='VDD/2' rise=1 targ V(out) val='VDD/2' rise=1
    .measure tran tf
149
    .measure tran tpdr trig V(in1) val='VDD/2'
150
    .measure tran tpdf trig V(in8) val='VDD/2'
                                                       fall=1 targ V(out) val='VDD/2'
    .measure tran tpd param='(tpdr+tpdf)/2'
152
153
    * power dissipation of the AND8 gate
    .inc ./and8_test_pow.inc
155
156
157
    .end
```

B.7.4. AND8C

Listing 21: and8c.inc

```
* ------
  * Circuit : CMOS AND8 Gate Type C
  * Description: AND4B * 2 + AND2 * 1
  * Author
              : Wuqiong Zhao (me@wqzhao.org)
5
  * Date
             : 2023-06-02
  * License
             : MIT
  * ______
8
  * The following warning may be raised:
10
11
      Warning: redefinition of .subckt nand2, ignored
12
13
      Warning: redefinition of .subckt nand, ignored
  * This is due to recursive .inc commands load the same .subckt.
15
  * You can safely ignore these 2 warnings.
16
17
   .subckt AND8C gnd i1 i2 i3 i4 i5 i6 i7 i8 o vdd
18
19
    * src gate drain body type
    XAND4B_1 gnd i1 i2 i3 i4 t1 vdd AND4B
20
    \tt XAND4B\_2 \ gnd \ i5 \ i6 \ i7 \ i8 \ t2 \ vdd \ AND4B
21
    XAND2
            gnd t1 t2
                        o vdd AND2
   .ends AND8C
23
24
  .inc ./and4b.inc
  .inc ./and2.inc
26
```

Listing 22: and8c.cir

```
*-----
   * Circuit : CMOS AND8 Gate Type C
  * Description: AND4B * 2 + AND2 * 1
  * Author : Wuqiong Zhao (me@wqzhao.org)
* Date : 2023-06-02
* License : MIT
6
   * ------
10 .title CMOS AND8C
11
   * Parameters and Model
12
13 * -
   .param VDD='1.0V'
14
   .temp 25
15
  .inc ./FreePDK45/ff.inc
17
  * Supply Voltage Source
18
19
20 Vdd vdd gnd VDD
21
  * AND8B
22
23
   XAND8 gnd in1 in2 in3 in4 in5 in6 in7 in8 out vdd AND8C
   .inc ./and8c.inc
25
26
27
   * Test Circuit
28
29 XTest gnd
30 + ii1 ii2 ii3 ii4 ii5 ii6 ii7 ii8
31 + in1 in2 in3 in4 in5 in6 in7 in8
        out vdd
                            INV2_TEST
   .inc ./and8_test_inv2.inc
33
34
  * Input Signals
35
36
   Vii1 ii1 gnd PWL
37
38 + (
39 + 0.0ns 0V
40
      0.9ns 0V
   + 1.1ns VDD
41
42 + 3.0ns VDD
43
44
45 Vii2 ii2 gnd PWL
46
  + (
      0.0ns 0V
47
48
   + 0.9ns 0V
      1.1ns VDD
49
       3.0ns VDD
50
  +
  + )
52
Vii3 ii3 gnd PWL
54
  + (
  + 0.0ns 0V
55
56
      0.9ns 0V
     1.1ns VDD
57
     3.0ns VDD
58 +
59
  + )
60
61
  Vii4 ii4 gnd PWL
  + (
62
  + 0.0ns 0V
63
       0.9ns 0V
      1.1ns VDD
65
       3.0ns VDD
66
67 + )
68
69 Vii5 ii5 gnd PWL
70 + (
71 + 0.0ns 0V
72 + 0.9ns 0V
73 + 1.1ns VDD
```

```
74 + 3.0ns VDD
75
    + )
76
   Vii6 ii6 gnd PWL
77
78
79
        0.0ns 0V
        0.9ns 0V
80
81
        1.1ns VDD
        3.0ns VDD
82
83 + )
84
    Vii7 ii7 gnd PWL
85
86 + (
87
        0.0ns 0V
        0.9ns 0V
88
    + 1.1ns VDD
       3.0ns VDD
90
91
    + )
    Vii8 ii8 gnd PWL
93
94
    + (
        0.0ns 0V
95
96
        0.9ns 0V
97
        1.1ns VDD
       1.9ns VDD
98
99
       2.1ns 0V
100
        3.0ns 0V
    + )
101
102
    * Analysis
103
104
    .ic V(out)=0
    .tran 0.005ns 3ns
106
107
108
    .control
    run
109
      * >>>> plot >>>>>
110
111
      set xgridwidth = 2
      set xbrushwidth = 3
112
      * "svgwidth", "svgheight", "svgfont-size", "svgfont-width", "svguse-color",
113

→ "svgstroke-width", "svggrid-width",
114
      set svg_intopts = ( 1024 256 16 0 1 2 0 )
      * "svgbackground", "svgfont-family", "svgfont"
115
      setcs svg_stropts = ( white Arial Arial )
116
117
      set hcopydevtype = svg
                    = black
= red
      set color1
118
      set color2
119
120
      set color3
                       = blue
121
      set color4
                       = green
122
      hardcopy fig/plot_and8c_t.svg
123
      + in1 out in8
124
      + title 'CMOS AND8c'
125
      + xlabel 't'
126
      + ylabel 'Voltage'
+ ylimit -0.2 1.2
127
128
      + ydelta 0.5
129
130
      * for MS Windows, using Edge
131
      if $oscompiled = 1 | $oscompiled = 8
132
133
        shell Start fig/plot_and8c_t.svg
134
      else
        if $oscompiled = 7
135
136
          * macOS (using Safari, no need to install X11)
          shell open -a safari fig/plot_and8c_t.svg &
137
138
139
          * for CYGWIN, Linux, using feh and X11
          shell feh --magick-timeout 1 fig/plot_and8c_t.svg &
140
141
        end
142
      * <<<<< plot <<<<
143
144
145
```

```
* Measurement
146
147
    .measure tran tr trig V(\text{out}) val='VDD*0.1' rise=1 targ V(\text{out}) val='VDD*0.9' rise=1 .measure tran tf trig V(\text{out}) val='VDD*0.9' fall=1 targ V(\text{out}) val='VDD*0.1' fall=1
148
149
     .measure tran tpdr trig V(in1) val='VDD/2' rise=1 targ V(out) val='VDD/2'
    .measure tran tpdf trig V(in8) val='VDD/2'
                                                               fall=1 targ V(out) val='VDD/2' fall=1
151
152
    .measure tran tpd param='(tpdr+tpdf)/2'
153
    * power dissipation of the AND8 gate
154
155
    .inc ./and8_test_pow.inc
156
157
     . end
```

B.8. Clock Controlled SR Latch

Listing 23: SR_latch_clk.inc

```
* ------
  * Circuit : Clock Controlled SR Latch
2
  \star Description: 2 PMOS + 6 NMOS
             : Wuqiong Zhao (me@wqzhao.org)
  * Author
5
  * Date : 2023-06-03
* License : MIT
  * -----
8
  * .param WL = 5
10
11
  .subckt SR_LATCH_CLK gnd s r clk q qn vdd
    * src gate drain body type
12
           q gnd gnd NMOS_VTL W=
    M1 qn
                                      90nm L=45nm
13
    M2 qn
               vdd vdd PMOS_VTL W=
                                    270nm L=45nm
14
           q
              gnd gnd NMOS_VTL W=
vdd vdd PMOS_VTL W=
    M3 q
                                      90nm L=45nm
15
           qn
                                      270nm I = 45nm
16
    M4 q
           qn
    M5 ts
               gnd gnd NMOS_VTL W={WL*45nm} L=45nm
17
          clk ts
               ts gnd NMOS_VTL W={WL*45nm} L=45nm
gnd gnd NMOS_VTL W={WL*45nm} L=45nm
    M6 qn
18
    M7 tr
19
           r
    M8 q
           clk tr
                    gnd NMOS_VTL W={WL*45nm} L=45nm
  .ends SR_LATCH_CLK
```

Listing 24: SR_latch_clk.cir

```
* ------
1
   * Circuit : Clock Controlled SR Latch
  * Description: 2 PMOS + 6 NMOS
3
              : Wuqiong Zhao (me@wqzhao.org)
  * Date
             : 2023-06-03
6
  * License
             : MIT
8
  .title CMOS Clock Controlled SR Latch
11
  * Parameters and Model
12
13
  .param VDD='1.0V'
14
15
   .temp 27
  .inc ./FreePDK45/ff.inc
16
17
  * Supply Voltage Source
19
20 Vdd vdd gnd VDD
  * Clock Controlled SR Latch
22
24
   .inc ./SR_latch_clk.inc
  XLatch gnd s r clk q qn vdd SR_LATCH_CLK
25
27
  * Input Signals
28
```

```
Vs s gnd PWL
29
30
        0.0ns 0V
31
        0.9ns 0V
32
33
        1.1ns VDD
        1.9ns VDD
34
35
    +
        2.1ns 0V
36
        3.0ns 0V
    + )
37
38
    Vr r gnd PWL
39
40
   + (
41
        0.0ns 0V
        2.4ns 0V
42
        2.6ns VDD
43
        3.5ns VDD
   + )
45
46
    Vclk clk gnd PWL
47
48
   + (
        0.0ns VDD
49
        3.0ns VDD
50
51
   + )
    * Analysis
53
55
    .ic
            V(qn) = VDD
          V(q) = 0V
    .ic
56
57
    .tran 0.005ns 3ns
58
    .param WL = 10
59
    .probe V(qn)
61
62
    .control
    set xgridwidth = 2
63
      set xbrushwidth = 3
64
      * "svgwidth", "svgheight", "svgfont-size", "svgfont-width", "svguse-color",
65

→ "svgstroke-width", "svggrid-width",
      set svg_intopts = ( 1024 256 16 0 1 2 0 )
* "svgbackground", "svgfont-family", "svgfont"
66
67
      setcs svg_stropts = ( white Arial Arial )
68
69
      set hcopydevtype = svg
      set color1
70
71
      * Sweep Parameters
72
      foreach x 2 4 4.2 4.5 5 6 8
73
       alterparam WL = x
74
75
       reset
        run
76
77
      end
78
      let WL2 = tran1.V(qn)
let WL4 = tran2.V(qn)
79
80
      let WL4.2 = tran3.V(qn)
81
      let WL4.5 = tran4.V(qn)
82
83
      let WL5
                = tran5.V(qn)
      let WL6
               = tran6.V(qn)
84
      let WL8 = tran7.V(qn)
85
86
      hardcopy fig/plot_sr_latch_wl_t.svg
87
88
      + s r WL2 WL4 WL4.2 WL4.5 WL5 WL6 WL8
      + title 'CMOS Clock Controlled SR Latch With Different \mbox{W/L'}
89
      + xlabel 't'
90
      + ylabel 'Voltage'
91
      + ylimit 0 1
92
93
      * for MS Windows, using Edge
if $oscompiled = 1 | $oscompiled = 8
95
96
        shell Start fig/plot_sr_latch_wl_t.svg
      else
97
       if $oscompiled = 7
98
99
          * macOS (using Safari, no need to install X11)
        shell open -a safari fig/plot_sr_latch_wl_t.svg &
100
```

B. Source Code B.9. MISC

```
else
101
          * for CYGWIN, Linux, using feh and X11
102
           shell feh --magick-timeout 1 fig/plot_sr_latch_wl_t.svg &
103
        end
104
105
      end
106
      * One example that This SR Latch will work.
107
108
      alterparam WL = 6
      reset
109
110
      run
      set color2
111
                         = red
      set color3
                        = blue
112
113
      set color4
                        = green
      set color5
                         = brown
114
115
116
      hardcopy fig/plot_sr_latch_t.svg
      + qn q s r
+ title 'CMOS Clock Controlled SR Latch'
117
118
      + xlabel 't'
119
      + ylabel 'Voltage'
120
      + ylimit 0 1
121
122
      \ensuremath{^{*}} for MS Windows, using Edge
123
124
      if $oscompiled = 1 | $oscompiled = 8
       shell Start fig/plot_sr_latch_t.svg
125
126
      else
127
        if $oscompiled = 7
           * macOS (using Safari, no need to install X11)
128
129
           shell open -a safari fig/plot_sr_latch_t.svg &
130
          * for CYGWIN, Linux, using feh and X11
131
           shell feh --magick-timeout 1 fig/plot_sr_latch_t.svg &
        end
133
134
      end
    .endc
135
136
137
    .end
```

B.9. MISC

Listing 25: .spiceinit

```
* Compatibility with HSPICE
set ngbehavior=hs
```